NAG Library Routine Document

C09AAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C09AAF returns the details of the chosen discrete wavelet filter. For a chosen mother wavelet, discrete wavelet transform type (single-level or multi-level DWT) and end extension method, this routine returns the number of levels of resolution (appropriate to a multi-level transform), the filter length, and, the number of approximation and detail coefficients, or the total number of coefficients (for single-level or multi-level DWTs respectively). This routine must be called before any of the transform routines in this chapter.

2 Specification

```
SUBROUTINE C09AAF(WAVNAM, WTRANS, MODE, N, NWL, NF, NWC, ICOMM, IFAIL)
INTEGER           N, NWL, NF, NWC, ICOMM(100), IFAIL
CHARACTER*(*)     WAVNAM
CHARACTER*1       WTRANS, MODE
```

3 Description

One-dimensional discrete wavelet transforms (DWT) are characterised by the mother wavelet, the end extension method and whether multiresolution analysis is to be performed. For the selected combination of choices for these three characteristics, and for a given length (n) of data array, C09AAF returns the dimension details for the transform determined by this combination. The dimension details are: n_l, the number of levels of resolution that would be computed were a multi-level DWT applied, n_f, the filter length, and n_c the number of approximation and detail coefficients for single-level DWTs and the total number of coefficients for multi-level DWTs that would be generated by the transform. These values are also stored in the communication array ICOMM, as are the input choices, so that they may be conveniently communicated to the transform routines in this chapter.

4 References

None.

5 Parameters

1: WAVNAM – CHARACTER*(*) *Input*

On entry: the name of the mother wavelet. See the C09 Chapter Introduction for details.

WAVNAM = 'HAAR'
Haar wavelet.

WAVNAM = 'DB***n***', where ***n*** $= 2, 3, \ldots, 10$
Daubechies wavelet with ***n*** vanishing moments (2***n*** coefficients). For example, WAVNAM = 'DB4' is the name for the Daubechies wavelet with 4 vanishing moments (8 coefficients).

WAVNAM = 'BIOR***x.y***', where ***x.y*** can be one of 1.1, 1.3, 1.5, 2.2, 2.4, 2.6, 2.8, 3.1, 3.3, 3.5 or 3.7

Biorthogonal wavelet of order ***x.y***. For example WAVNAM = 'BIOR3.1' is the name for the Biorthogonal wavelet of order 3.1.

Constraint: WAVNAM = 'HAAR', 'DB2', 'DB3', 'DB4', 'DB5', 'DB6', 'DB7', 'DB8', 'DB9', 'DB10', 'BIOR1.1', 'BIOR1.3', 'BIOR1.5', 'BIOR2.2', 'BIOR2.4', 'BIOR2.6', 'BIOR2.8', 'BIOR3.1', 'BIOR3.3', 'BIOR3.5' or 'BIOR3.7'.

2: WTRANS – CHARACTER*1 *Input*

On entry: the type of discrete wavelet transform that is to be applied.

WTRANS = 'S'
Single-level decomposition or reconstruction by discrete wavelet transform.

WTRANS = 'M'
Multi-level resolution, by a multi-level DWT or its inverse.

Constraint: WTRANS = 'S' or 'M'.

3: MODE – CHARACTER*1 *Input*

On entry: the end extension method.

MODE = 'P'
Periodic end extension.

MODE = 'H'
Half-point symmetric end extension.

MODE = 'W'
Whole-point symmetric end extension.

MODE = 'Z'
Zero end extension.

Constraints:

if WTRANS = 'S', MODE = 'P', 'H', 'W' or 'Z';
if WTRANS = 'M', MODE = 'H', 'W' or 'Z'.

4: N – INTEGER *Input*

On entry: the number of elements, n, in the data array x.

Constraint: N ≥ 2.

5: NWL – INTEGER *Output*

On exit: the maximum number of levels of resolution, n_l, that can be computed when a multi-level discrete wavelet transform is applied. It is such that $2^{n_l} \leq n < 2^{n_l+1}$, for n_l an integer.

6: NF – INTEGER *Output*

On exit: the filter length, n_f, for the supplied mother wavelet. This is used to determine the number of approximation coefficients that are appropriate to each level of a multi-level DWT.

7: NWC – INTEGER *Output*

On exit: for a single-level transform (WTRANS = 'S'), the number of approximation coefficients that would be generated for the given problem size, mother wavelet, extension method and type of transform; this is also the corresponding number of detail coefficients. For a multi-level transform (WTRANS = 'M') the total number of coefficients that would be generated over all levels of resolution.

8: ICOMM(100) – INTEGER array *Communication Array*

On exit: contains details of the wavelet transform and the problem dimension which is to be communicated to the discrete transform routines in this chapter.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, WAVNAM ≠ 'HAAR', 'DB2', 'DB3', 'DB4', 'DB5', 'DB6', 'DB7', 'DB8', 'DB9', 'DB10', 'BIOR1.1', 'BIOR1.3', 'BIOR1.5', 'BIOR2.2', 'BIOR2.4', 'BIOR2.6', 'BIOR2.8', 'BIOR3.1', 'BIOR3.3', 'BIOR3.5' or 'BIOR3.7'.

IFAIL = 2

On entry, WTRANS ≠ 'S' or 'M'.

IFAIL = 3

On entry, MODE ≠ 'P', 'H', 'W' or 'Z',
or MODE = 'P' and WTRANS = 'M'.

IFAIL = 4

On entry, N < 2.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example computes the one-dimensional multi-level resolution for 8 values by a discrete wavelet transform using the Haar wavelet with zero end extensions. The length of the wavelet filter, the number of levels of resolution, the number of approximation coefficients at each level and the total number of wavelet coefficients are printed.

9.1 Program Text

```
*     C09AAF Example Program Text
*     Mark 22 Release. NAG Copyright 2007.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NWCMAX, NWLMAX
      PARAMETER        (NMAX=850,NWCMAX=850,NWLMAX=20)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, N, NF, NNZ, NWC, NWL, NY
      CHARACTER*10     MODE, WAVNAM, WTRANS
*     .. Local Arrays ..
      DOUBLE PRECISION C(NWCMAX), X(NMAX), Y(NMAX)
      INTEGER          DWTLEV(NWLMAX), ICOMM(100)
*     .. External Subroutines ..
      EXTERNAL         C09AAF, C09CCF, C09CDF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C09AAF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
*     Read problem parameters
      READ (NIN,*) N
      READ (NIN,*) WAVNAM, MODE
*
      IF (N.LE.NMAX .AND. N.GE.2) THEN
         WRITE (NOUT,99999) WAVNAM, MODE, N
*
*        Read data array and write it out
*
         READ (NIN,*) (X(I),I=1,N)
*
         WRITE (NOUT,*) ' Input Data           X :'
         WRITE (NOUT,99998) (X(I),I=1,N)
*
*        Query wavelet filter dimensions
*        For Multi-Resolution Analysis, decomposition, WTRANS = 'M'
*
         IFAIL = 1
         WTRANS = 'Multilevel'
         CALL C09AAF(WAVNAM,WTRANS,MODE,N,NWL,NF,NWC,ICOMM,IFAIL)
*
         IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99997) IFAIL
            GO TO 40
         END IF
*
*        Perform Discrete Wavelet transform
*
         IFAIL = 1
         CALL C09CCF(N,X,NWCMAX,C,NWL,DWTLEV,ICOMM,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99996) NF
            WRITE (NOUT,99995) NWL
            WRITE (NOUT,99994)
            WRITE (NOUT,99993) (DWTLEV(I),I=1,NWL+1)
            WRITE (NOUT,99992) NWC
            NNZ = 0
            DO 20 I = 1, NWL + 1
               NNZ = NNZ + DWTLEV(I)
   20       CONTINUE
            WRITE (NOUT,*)
            WRITE (NOUT,99991)
            WRITE (NOUT,99998) (C(I),I=1,NNZ)
*
*           Reconstruct original data
*
            NY = N
```

```
            IFAIL = 1
            CALL C09CDF(NWL,NWCMAX,C,NY,Y,ICOMM,IFAIL)
*
            IF (IFAIL.NE.0) THEN
               WRITE (NOUT,99990) IFAIL
            ELSE
               WRITE (NOUT,*)
               WRITE (NOUT,99989)
               WRITE (NOUT,99998) (Y(I),I=1,NY)
            END IF
         ELSE
            WRITE (NOUT,99988) IFAIL
         END IF
*
   40 END IF
*
99999 FORMAT (1X,' Parameters read from file :: ',/'     Wavelet : ',
     +       A10,' End mode : ',A10,' N = ',I10)
99998 FORMAT (8(F8.3,1X),:)
99997 FORMAT (1X,' ** C09AAF returned with IFAIL = ',I5)
99996 FORMAT (1X,' Length of wavelet filter :             ',I10)
99995 FORMAT (1X,' Number of Levels :                     ',I10)
99994 FORMAT (1X,' Number of coefficients in each level : ')
99993 FORMAT (16X,8(I8,1X),:)
99992 FORMAT (1X,' Total number of wavelet coefficients : ',I10)
99991 FORMAT (1X,' Wavelet coefficients C : ')
99990 FORMAT (1X,' ** C09CDF returned with IFAIL = ',I5)
99989 FORMAT (1X,' Reconstruction       Y : ')
99988 FORMAT (1X,' ** C09CCF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
C09AAF Example Program Data
8               : N
Haar  Zero      : WAVNAM, MODE
 2.0
 5.0
 8.0
 9.0
 7.0
 4.0
-1.0
 1.0            : X(1:N)
```

9.3 Program Results

```
 C09AAF Example Program Results

  Parameters read from file ::
     Wavelet : Haar       End mode : Zero       N =          8
  Input Data           X :
   2.000    5.000    8.000    9.000    7.000    4.000   -1.000    1.000

  Length of wavelet filter :                       2
  Number of Levels :                               3
  Number of coefficients in each level :
                       1        1        2        4
  Total number of wavelet coefficients :           8

  Wavelet coefficients C :
  12.374    4.596   -5.000    5.500   -2.121   -0.707    2.121   -1.414

  Reconstruction       Y :
   2.000    5.000    8.000    9.000    7.000    4.000   -1.000    1.000
```